Template for *Mathematica* Assignment 3

1) We consider the space curve given by

$$\vec{r}(t) = \langle (3\cos(3t) + 5\sin(t))^2,\ 3\cos(3t) - \cos(t),\ (\sin(2t) + 3\sin(t))^2 \rangle$$

Clearly, this function is periodic with a period of 2π. We can plot this curve using ParametricPlot3D.

```
p1 = ParametricPlot3D[{(3 * Cos[3 t] + 5 * Sin[t]) ^ 2, 3 * Cos[3 t] - Cos[t], (Sin[2 t] + 3 Sin[t]) ^ 2},
  {t, 0, 2 * Pi}, AxesLabel → {"x", "y", "z"}]
```



For the arclength of this curve, we could try to get an exact answer:

```
ArcLength[{(3 * Cos[3 t] + 5 * Sin[t]) ^ 2, 3 * Cos[3 t] - Cos[t], (Sin[2 t] + 3 Sin[t]) ^ 2}, {t, 0, 2 * Pi}]
$Aborted
```

Unfortunately, this curve is ugly enough that even *Mathematica* has trouble giving an exact answer! Instead, we pass an option to the ArcLength function telling it to integrate numerically.

```
In[4]:= ArcLength[{(3 * Cos[3 t] + 5 * Sin[t]) ^ 2, 3 * Cos[3 t] - Cos[t], (Sin[2 t] + 3 Sin[t]) ^ 2},
  {t, 0, 2 * Pi}, WorkingPrecision → 10]
Out[4]= 278.7806979
```

Note that we set the WorkingPrecision to 10 in order to get more decimal places for the answer. We could have asked for far more accuracy, but this will suffice for our purposes!

Next, we look at the curvature. As for many things we encounter in the course, *Mathematica* already has a built in function to compute the curvature of parametric curves:

```
FullSimplify[
 ArcCurvature[{(3 * Cos[3 t] + 5 * Sin[t])^2, 3 * Cos[3 t] - Cos[t], (Sin[2 t] + 3 Sin[t])^2}, t]]
```

```
√((20 964 240 Cos[t] + 8 711 694 Cos[2 t] + 24 438 768 Cos[3 t] - 17 857 332 Cos[4 t] -
    30 602 448 Cos[5 t] - 19 034 321 Cos[6 t] - 15 594 408 Cos[7 t] - 16 829 370 Cos[8 t] +
    4 428 552 Cos[9 t] + 4 145 535 Cos[10 t] - 2 (-1 619 856 Cos[11 t] - 6 092 050 Cos[12 t] +
       3 (-7 526 179 - 521 316 Cos[13 t] + 188 703 Cos[14 t] - 78 408 Cos[15 t] + 29 844 Cos[16 t] +
          221 616 Cos[17 t] + 212 139 Cos[18 t] + 26 244 Cos[19 t] + 1944 Cos[20 t] + 509 400 Sin[t] +
          9 742 260 Sin[2 t] + 4 248 000 Sin[3 t] + 3 504 480 Sin[4 t] + 4 864 320 Sin[5 t] +
          288 940 Sin[6 t] - 3 527 640 Sin[7 t] - 1 275 360 Sin[8 t] - 1 312 920 Sin[9 t] -
          3 947 220 Sin[10 t] - 518 400 Sin[11 t] + 506 560 Sin[12 t] - 195 840 Sin[13 t] +
          247 860 Sin[14 t] + 456 840 Sin[15 t] + 305 640 Sin[16 t] + 19 440 Sin[17 t]))) /
  (6111 + 144 Cos[t] - 3646 Cos[2 t] + 42 Cos[3 t] - 1084 Cos[4 t] - 150 Cos[5 t] -
     3798 Cos[6 t] - 36 Cos[7 t] + 4946 Cos[8 t] - 729 Cos[12 t] - 6240 Sin[2 t] +
     120 Sin[4 t] + 3000 Sin[6 t] + 1620 Sin[8 t] - 3240 Sin[10 t])^3)
```

Even after simplifying (note the FullSimply command), this is an unholy mess! Let's get a plot for it.

```
Plot[%8, {t, 0, 2 * Pi}, AxesLabel → {"t", "κ"}]
```



This is another instance where *Mathematica* does not do the best job of choosing a viewing window! Since we want to find the point on the curve where the curvature is maximum, we zoom out on the vertical axis.

```
Plot[%8, {t, 0, 2*Pi}, PlotRange → {{0, 2*Pi}, {0, 50}}, AxesLabel → {"t", "κ"}]
```



It looks as though the *t*-value where the curve has maximum curvature is somewhere in the range $5.1 < t < 5.2$. To get a better sense of where this maximum is located, we take the derivative of the curvature and set it equal to zero. Obviously, this is not something we want to do by hand!

```
FindRoot[D[%8, t], {t, 5.1, 5.2}, Method → "Brent", WorkingPrecision → 30]
{t → 5.15613391545140921758304370871}
```

Note, the method called in the FindRoot command is not named for me! It is a procedure developed by Richard Brent in 1973. The two numbers 5.1 and 5.2 in the command give bounds on where the root we want to find lies on the *t* axis. Setting the WorkingPrecision to 30 is definitely overkill here, but *Mathematica* has no problem complying!

In order to mark the point on the curve with the highest curvature, we use the Graphics3D command with the option Sphere (there is an option called Point, but it tends to give bad results in 3D images):

```
p2 = Graphics3D[{Red, Sphere[
      {(3 * Cos[3 t] + 5 * Sin[t]) ^ 2, 3 * Cos[3 t] - Cos[t], (Sin[2 t] + 3 Sin[t]) ^ 2} /. %58, 0.3]}];
Show[
 p1,
 p2]
```



Certainly the point marked by the red sphere looks like it has the sharpest curvature! Be careful about scaling though! If you have different scales on the different axes, the curvature at various points may appear to be larger or smaller than they actually are!